

October, 2003

Advisor Answers

Know Whether a Report was Printed

VFP 8

Q: Is there a way to tell that a report preview being printed? If you use REPORT FORM PREVIEW, how can you tell when the user clicks the Print button?

-Joey Carroll (via CompuServe.COM)

A: Before VFP 8, you had to jump through all sorts of hoops to get this information. VFP 8 adds a new function, SYS(2040), that tells you when a report is running whether it's being sent to the printer or is displaying as a preview. You can use this information in a variety of ways.

SYS(2040) itself is pretty simple. It returns a single character value. If you call it from anywhere other than a report, it returns "0". From a preview, the return value is "1" and from a report sent to the printer, it returns "2".

One way to use SYS(2040) is to have some items appear only on printed reports or only in previews. For example, if you set up a watermark on a report, you might choose to show it only in the printed report and not in the preview, since it could make it hard to read a preview. To use SYS(2040) this way, specify something like the following as the Print When expression for the item:

```
SYS(2040)<>"1"
```

To include an item in a preview, but not in a printed report, just reverse the expression:

```
SYS(2040)="1"
```

You can do a lot more with SYS(2040) if you want. For example, consider the case of an invoice. The first time you print it, nothing special is needed, but subsequent printings should include the word "Copy." To do that, you need to know when you print it the first time. One way to handle that is to record the information when the invoice actually prints.

This month's Professional Resource CD includes an example report, Invoice.FRX, that is grouped by the primary key of the Invoice table (also included on the PRD). The group footer has this On Exit expression:

```
IIF(sys(2040)="2",UpdateInvoice(),.t.)
```

That is, when the report is sent to the printer, a function, UpdateInvoice() is called for each record. The function updates the record for the current invoice. Here's the code (found on this month's PRD), which is pretty simple for this example:

```
* UpdateInvoice.PRG
* Update current record in Invoice table
* indicate invoice was printed
TRY
  IF NOT Invoice.lPrinted
    REPLACE lPrinted WITH .T. in Invoice
    REPLACE dInvoice WITH DATE() IN Invoice
  ENDIF
CATCH
  * Do nothing
ENDTRY
```

The group footer band of the report has the following item:

```
iif(lPrinted, "Copy", "")
```

The invoice date printed on the report also uses the information that's updated by UpdateInvoice(). It contains this expression:

```
EVL(dInvoice,DATE())
```

EVL() is a new function in VFP 8; it's analogous to NVL(). It evaluates the first expression you pass it. If that expression is not empty, it returns the value of the first expression. If the first expression is empty, the function returns the value of the second expression. In other words, the expression above is equivalent to this one:

```
IIF(EMPTY(dInvoice), DATE(), dInvoice)
```

As designed, previews of the invoice show the invoice date as the current date until it's printed for the first time. At that point, the invoice date is stored in the record. From then on, the stored data is displayed.

I should probably point out a couple of weaknesses with this approach. First, it's possible that a user might cancel the report after some invoices have been processed, but before it's finished. That would

cause the invoices already processed to be marked as printed, even though they weren't. One solution is to wrap the report in a transaction. Issue BEGIN TRANSACTION before the report runs. Set a flag in the Summary band to indicate that the report has completed. After the report, check the flag and if it's been set, issue END TRANSACTION; if not, ROLLBACK. The code would look something like this:

```
* Prepare data for invoice report
* Set flag
lCompleted = .F.
* Start transaction and run report
BEGIN TRANSACTION
REPORT FORM Invoice TO PRINT
* Check flag
IF lCompleted
    END TRANSACTION
ELSE
    ROLLBACK
ENDIF
```

The second problem is that SYS(2040) tells you only whether the report was sent to the print spooler, not whether it actually printed. If there's a problem with the printer or the user cancels the report in the Print Manager, the report may not actually get printed.

-Tamar